
Detection and quantification of intracellular signalling using FRET-based 

biosensors and high content imaging. 

Michelle L Halls, Daniel P Poole, Andrew M Ellisdon, Cameron J Nowell and Meritxell 

Canals 

Summary  

Förster Resonance Energy Transfer (FRET) biosensors represent invaluable tools to 

detect the spatiotemporal context of second messenger production and intracellular 

signalling that cannot be attained using traditional methods. Here, we describe a detailed 

protocol for the use of high content imaging in combination with FRET biosensors to 

assess second messenger production and intracellular signalling in a time-effective 

manner. We use four different FRET biosensors to measure cAMP levels, kinase (ERK 

and PKC) and GTPase activity. Importantly, we provide the protocols to express and 

measure these sensors in a variety of model cell lines and primary dorsal root ganglia 

neurons.  

Key Words: cell signalling, Förster Resonance Energy Transfer, biosensors, high content 

imaging, extracellular regulated kinase, protein kinase C, cyclic AMP, Rac1. 

1. Introduction  

The study of complex signalling systems in living cells is now possible due to a growing 

range of biosensors designed to alter Förster Resonance Energy Transfer (FRET) signals 

upon activation of specific signalling pathways [1]. These FRET biosensors represent a 

major advance over traditional population-based assays, as they allow visualisation of the 

spatial and temporal dynamics of intracellular signalling in real time and in live cells. Their 

use is further facilitated by the fact they are genetically encoded and require very simple 

optics and software for analysis. For example, FRET sensors have been described that 

detect changes in cAMP levels, specific kinase activity or GTPase activity [2-7]. 

Importantly, these biosensors can be targeted to subcellular compartments (nucleus, 

mitochondria, cytosol, plasma membrane), allowing the detection of distinct signalling 

events with subcellular resolution (Fig 1).  

[Fig 1 near here] 

The use of FRET biosensors to visualise signalling at the single cell level also provides the 

opportunity to discriminate the responses of individual cells within a mixed cell population 

(for example in primary neuronal cultures). Therefore the use of such sensors provides an 

additional level of information that would otherwise be lost in population-based signalling 

assays, where the total response reflects a composite of many individual responses (Fig 

2).  

[Fig 2 near here] 

Up until recently, FRET reporters were typically used in combination with imaging systems 

that have very limited throughput. Although these systems are highly sensitive, their use of 

entails lengthy acquisition sessions. Recent advances in high content image acquisition 

and analysis provide a unique opportunity to assess second messenger production and 

intracellular signalling in a more time-effective manner. Here, we describe protocols for the 

use of high content imaging to investigate intracellular signalling in model cell lines and 

primary neuronal cultures using FRET biosensors. We also provide macros for semi-

automated analysis of the FRET measurements, and additional details that should be 

considered at each stage of the experimental design.   

2. Materials 

2.1. Cell culture and transfection 

1. Dulbecco’s Modification of Eagle’s Medium (DMEM). 

2. Foetal Calf Serum (FCS). 

3. Roswell Park Memorial Institute-1640 medium (RPMI). 

4. N1 neuronal supplement: 0.5 mg/mL recombinant human insulin, 0.5 mg/mL human 

transferrin (partially iron-saturated), 0.5 μg/mL sodium selenite, 1.6 mg/mL 

putrescine, and 0.73 μg/mL progesterone. 

5. Ca2+/Mg2+-free Hank’s Balanced Salt Solution (CMF-HBSS): 137 mM NaCl, 5.3 mM 

KCl, 0.44 mM KH2PO4, 4.1 mM NaHCO3, 0.34 mM Na2HPO4, pH 7.4. Autoclave 

and store at 4°C. Add glucose at 1 g/L prior to use. 

6. Cell line growth media: DMEM, 5% FCS (see Note 1). 

7. Primary neuronal culture growth media: DMEM, 10% FCS, 1% antibiotic-

antimycotic solution, 1% N1 neuronal supplement (see Note 2).  

8. Collagenase Type IV. 

9. Dispase II. 

10. Digestion buffer: 2 mg/mL Collagenase Type IV, 2 mg/mL Dispase II in CMF-HBSS. 

Make 500 µL per mouse in 1.6 mL microtubes. 

11. Sterile Phosphate Buffered Saline (PBS) Mg2+/Ca2+-free: 137 mM NaCl, 2.7 mM 

KCl, 1.5 mM KH2PO4, 8 mM Na2HPO4, pH 7.4. 

12. PBS/EDTA: sterile PBS with 0.5 mM EDTA. 

13. Laminin (1 mg/mL). 

14. Poly-D-Lysine (PDL). Prepare a stock solution of 1 mg/mL PDL in sterile ultra pure 

water. Filter sterilise, aliquot and store at -20°C (see Note 3). 

15. Dissection instruments: scissors (straight, ≥ 25 mm cutting edge), forceps (e.g. 

Dumont #5), spring scissors (e.g. Noyes 14 mm cutting edge, straight), spring 

scissors (e.g. Vannas 3 mm cutting edge, curved). 

16. Dissection microscope and light source. 

17. Nucleofection system and reagents. Lonza 4D-NucleofectorTM system and P3 

Primary Cell 4D- NucleofectorTMX Kit have been used for this protocol. 

18. Bunsen burner. 

19. Glass Pasteur pipettes and rubber bulb (see Note 4).  

20. Water bath. 

21. Incubator (37°C, 5% CO2/air). 

22. Cell culture hood. 

23. Black, optically clear PDL-coated 96-well plates (ViewPlate, PerkinElmer). For 

neuronal cultures, use a half-area black, optically clear 96-well plate to increase cell 

density. To coat plates, dilute PDL to 50 μg/mL in PBS and incubate wells with 50 

μL diluted PDL for 10 min. Rinse wells twice with 100 μL of PBS (see Notes 3 and 

5). 

24. 150 mM NaCl (filter sterilised). 

25. Polyethilenimine (PEI), linear molecular weight ~25000 Da. Prepare a stock solution 

of 1 mg/mL PEI in 150 mM NaCl. Filter sterilise, aliquot and store at -20°C (see 

Note 6). 

26. DNA constructs: receptor/protein of interest, FRET biosensors. A complete list of 

available FRET biosensors can be viewed at www.fret.lif.kyoto-u.ac.jp/e-

phogemon/unifret.htm. The FRET biosensors routinely used in our laboratories are 

detailed below (see Table 1). 

2.2. FRET detection and image analysis 



	

 6	

1. Hank’s Balanced Salt Solution (HBSS): 137 mM NaCl, 5.3 mM KCl, 1.3 mM CaCl2, 

0.41 mM MgSO4, 0.5 mM MgCl2, 0.44 mM KH2PO4, 4.1 mM NaHCO3, 0.34 mM 

Na2HPO4, pH 7.4. Autoclave and store at 4°C. Add glucose at 1 g/L prior to use. 

2. Stock solution of drugs and vehicle control. 

3. Stock solution of the positive control appropriate to the FRET biosensor to be used 

(see Table 2 and Note 7).  

4. GE Healthcare INCell 2000 version 4-10590 (INCell software version 6.01_21986) 

with FRET module (see Note 8). 

5. Image analysis software. The macros in this protocol use the FIJI distribution of 

ImageJ [8]. 

6. GraphPad Prism analysis software. 

3. Methods 

3.1. Cell culture 

3.1.1. Cell lines  

1. HEK293 cells are grown in DMEM, 5% FCS in culture flasks or dishes. Cells should 

be kept in incubators with a humidified atmosphere containing 5% CO2 at 37°C and 

passaged once they reach 80-90% confluency. 

2. MCF7 cells are grown in DMEM, 5% FCS and 10 μg/mL insulin in culture flasks or 

dishes. Cells should be kept in incubators with a humidified atmosphere containing 

5% CO2 at 37°C and passaged once they reach 80-90% confluency. 

3.1.2. Dorsal root ganglion neurons isolation and culture (see Note 9) 

This protocol has been optimised for use with dorsal root ganglion (DRG) neurons from 

adult mice (see Note 10) and utilises the Lonza 4D-Nucleofector™ system in combination 

with Nucleovette™ strips (a summary of the advantages and disadvantages of various 

delivery methods in primary neuronal cultures is provided by Zeitelhofer et al. [9]).  

1. Kill mouse using method approved by institutional animal ethics committee (e.g. 

cervical dislocation). 

2. Clean dorsal skin with 70% ethanol. Decapitate mouse and cut skin caudally along 

the vertebral column using scissors. 

3. Remove vertebral column starting from the cranial end. Lift as you cut and ensure 

that viscera are not attached. Trim skeletal muscle and ribs (see Note 11).  

4. Place tissue onto a petri dish. Open spinal canal from the cranial end by cutting 

halfway down the transverse face on either side. Open approximately 1 cm of the 

spinal column using the larger Noyes spring scissors (see Note 12).  

5. Harvest DRGs using the dissection microscope. Remove and discard the spinal 

cord. DRG should be visible in pockets on either side at the base of the vertebrae. 

Isolate the ganglia by grasping the associated root with forceps (#5) and pulling 

gently. This will move the ganglion slightly out of the pocket. Use the small curved 

Vannas spring scissors to cut the nerve located behind the ganglion while still 

pulling on the root. Place the extracted ganglion in a 35 mm diameter petri dish 

containing cold CMF-HBSS on ice. Harvest all DRG along the spine unless the 

experiment requires the collection of specific levels only. 

6. Once collected, trim any residual roots and attached connective tissue from the 

isolated ganglia. This will result in a cleaner neuronal culture. 

7. Place the cleaned DRG into a 1.6 mL microtube containing 500 μL pre-warmed 

digestion buffer. Incubate at 37°C in a water bath or incubator. 

8. After 30 min, triturate using the largest diameter Pasteur pipette (~5 times). Push 

the pipette tip to the bottom of the tube and take care not to lose any content from 

the tube. The solution should now appear cloudy. Return tube to 37°C. 

9. After 15 min, repeat the trituration step above using the medium diameter pipette. 

10. After 10 min, repeat the trituration step using the smallest diameter pipette 

11. Remove laminin solution from the PDL-coated cell culture plates (see Note 5) and 

allow the plates to air dry in the cell culture hood. 

12. Transfer digests to a 15 mL tube containing 5 mL DMEM and centrifuge (500 x g, 5 

min). Aspirate supernatant and resuspend the pellet in 5 mL DMEM. Repeat this 

washing step three times to remove any residual digestive enzymes.  

13. Proceed to Section 3.2.3. 

3.2. Transfection 

3.2.1. Introduction 

1. To investigate the signalling of heterologously expressed receptors, co-transfection 

of the receptor and FRET biosensor of interest is required. Alternatively, if the 

protein of interest is endogenously expressed, transfect only the FRET biosensor of 

interest. 

2. The transfection method depends on the cell line used for each study. Most model 

cell lines (including HEK293, CHO, COS, HeLa cells) can be transfected using 

inexpensive methods such as PEI or calcium phosphate. If these transfection 

methods are unsuccessful, more expensive methods such as Lipofectamine2000 or 

nucleofection may be required. 

3.2.2. Cell transfection using PEI 

1. Seed the cells into a 96-well black, optically clear PDL-coated plate in culture 

medium at a density which achieves 50% confluency in 24 h. 

2. Incubate the cells in a 37°C incubator in a humidified atmosphere containing 5% 

CO2 for 24 h. 

3. Prepare microtubes containing 50-100 ng of DNA (see Note 13) in sterile 150 mM 

NaCl to a total volume of 10 μL per well.  

4. In separate tubes, prepare a solution of PEI in sterile 150 mM NaCl to a final 

volume of 10 μL per well. The ratio of DNA:PEI should be optimised for each cell 

line, but an initial ratio of 1:6 is recommended. For example, 50 ng of DNA would 

require 300 ng of PEI for a 1:6 ratio. 

5. Add the diluted PEI solution (step 4) to the diluted DNA solution (step 3), mix 

thoroughly with a vortex, and incubate for 10-30 min at room temperature. 

6. Change the media of the cells. Add 200 μL of fresh DMEM with 5% FCS to each 

well. 

7. Add 20 μL/well of the DNA:PEI mix to the cells. 

8. Incubate the cells in a 37°C incubator in a humidified atmosphere containing 5% 

CO2. 

9. Experiments are performed 48 h after transfection following serum restriction of the 

cells by incubation overnight in DMEM with 0.5% FCS (see Notes 14 - 17) 

3.2.3. Cell transfection by nucleofection 

1. Following the final wash (see step 12 in Section 3.1.2.), resuspend pellet in 20 μL 

P3 solution containing 600 ng plasmid (see Note 18).  

2. Transfer digests from each mouse to individual wells in the Nucleocuvette™ strip 

and nucleofect (program DC104 according to manufacturer’s instructions). 

3. Immediately, add pre-warmed RPMI medium (150 μL) to fill each well. Place the 

strip in the incubator for 10 min to allow neurons to recover. 

4. Transfer solution to a 1.6 mL microtube and add complete media to bring the total 

volume to 100 μL per well (e.g. add an additional 430 μL medium to seed neurons 

into 6 wells of a half-area 96-well plate). Seed neurons into a PDL/laminin pre-

coated black, optically clear half-area 96-well plate, and incubate in a 37°C 

incubator in a humidified atmosphere containing 5% CO2.  

5. Experiments are performed after 48 h in vitro following serum restriction of the cells 

by incubation overnight in DMEM with 0.5% FCS. 

3.3. FRET detection using a high content imager 

This protocol is based on the use of the widefield high content imager INCell 2000. 

However, the steps described below should also be applicable to other high content 

imagers. 

3.3.1. Equilibration of cells 

1. Add 1 g/L glucose to stock HBSS and pre-warm to 37°C. 

2. Aspirate serum-restricted cell growth medium and gently add 160 μL pre-warmed 

complete HBSS to each well (see Note 19).  

3. If pre-treating cells with inhibitors, add 140 μL pre-warmed complete HBSS to each 

well, then 20 μL of inhibitor (diluted in HBSS) to the desired final concentration. 

4. Equilibrate cells in HBSS at 37°C without CO2 for 30 min (see Note 20). 

3.3.2. Instrument set up: generating an Acquisition Protocol using the GE INCell 2000 with 

FRET module 

1. Open a new Acquisition Protocol file, using the Protocol Designer window. 

2. Plate/Slide: select the appropriate plate template from the drop-down list (see Note 

21). 

3. Objective: select the 40x objective (see Note 22). 

4. Microscopy: select the number of wavelengths to be collected and the appropriate 

polychroic filter set (Quad); list and define the wavelengths in the order they are to 

be acquired (see Table 3 and Notes 23-25). 

5. Focus: select the ‘Refocus at each timepoint’ option. Focus should be optimised at 

the commencement of each experiment (see Section 3.3.3). 

6. Plate Heater: select the ‘Use plate heating’ option, and set to 37°C. 

7. Time Series: select the ‘Acquire time series’ option, and ‘look walk look’ mode 

(acquires images from each selected well within the specified time period). Specify 

the starting time point, the time interval (in seconds) and the number of time points 

to be acquired. For example, start at time 0, with interval of 60 seconds, and 

acquire 21 time points. This protocol will acquire an image every minute for 20 

minutes (see Note 26). 

3.3.3. Stimulation of cells and image acquisition 

1. Pre-heat GE INCell 2000 to 37°C, switch on the lamp, and open the FRET 

biosensor Acquisition Protocol (see Section 3.3.2). 

2. Highlight the wells to be acquired using the Plate/Slide Viewer, and select a region 

of interest (ROI, designated by the red square) within a well (see Notes 27 and 28). 

3. Define the optimal focus strategy for the acquisition. Within the Focus panel in the 

Protocol Designer, select the ‘Laser autofocus’ option then ‘Auto Offset’. Adjust the 

calculated offset and exposure time as required (see Note 29). 



	

 12	

4. Collect a ‘Baseline’ FRET time series: using an interval of 60 seconds, acquire 5 

time points. For time 0 select the brightfield wavelength only, for times 1-4 select 

FRET and donor wavelengths only. 

5. Stimulate the same cells with vehicle or ligand of interest, and collect a ‘Stimulated’ 

FRET time series: using an interval of 60 seconds, acquire 21 time points. Select 

FRET and donor wavelengths only (see Note 30). 

6. Stimulate the same cells with the appropriate positive control (see Section 2.2) for 

10 min. Collect a ‘Positive’ FRET time series: using an interval of 60 seconds, 

acquire 4 time points. Select FRET and donor wavelengths only (see Note 31). 

3.4. Image analysis 

3.4.1. Introduction 

Pre-processing and image analysis can be performed in many different ways using a large 

number of image analysis environments. For this protocol, pre-processing and image 

analysis is performed using three in-house macros that run in the FIJI distribution of 

ImageJ (see Notes 32 and 33). The first (Stack Creator v2.0) combines all image files into 

one hyperstack per field of view (i.e. the hyperstack contains the baseline, stimulated and 

positive control time points from both the donor and FRET channels) (see Note 34). Once 

the images are combined by the Stack Creator macro they can be processed by the two 

FRET Analysis macros. The first (FRET Analysis – Cell Markup v1.0) requires the user to 

manually select cells and define the background. The second (FRET Analysis – Batch 

Analyse v1.0) batch processes the hyperstacks and automatically calculates the 

background corrected FRET ratio, F/F0 and F/FMax values for all fields and timepoints. This 

macro also saves an overview image of each field as a record of the cells that were 

analysed. A breakdown of the macros is provided below. Final data is visualised using 

GraphPad Prism v6.0 (Fig. 3).  

[Link to macros here] 

3.4.2. Automated stacking and image alignment using the Stack Creator v2.0 macro  

1. Move all brightfield images from the baseline time series into a new directory. 

2. Open the Stack Creator v2.0 macro using FIJI, and press Run. 

3. Follow the prompts to choose the directories that contain the baseline, stimulated 

and positive control time series, and a new output directory in which to save the 

final hyperstacks. 

4. Define the settings for Stack Creator by entering the total number of wells scanned, 

the number of fields scanned per well, the well from which to start processing (see 

Note 35), and the number of baseline/stimulated/positive time points captured 

within each series. 

5. If the donor channel is listed first within the directory, select the check box. 

6. For automatic alignment of the images within the hyperstack, select the check box. 

7. Select OK. 

3.4.3. Defining ROIs using the FRET Analysis – Cell Markup v1.0 macro 

1. Open the FRET Analysis – Cell Markup v1.0 macro using FIJI, and press Run. 

2. Follow the prompt to choose the directory that contains the hyperstack generated 

by the Stack Creator macro. 

3. Follow the prompt to define the background by drawing a circle in an area without 

any cells, and press OK. 

4. Follow the prompt to mark out all cells of interest (see Notes 36-38), then press OK. 

5. Repeat for all hyperstacks within the selected directory. 

3.4.4. Data analysis using the FRET Analysis – Batch Analyse v1.0 macro 

1. Open the FRET Analysis – Batch Analyse v1.0 macro using FIJI, and press Run. 

2. Follow the prompt to choose the directory that contains the hyperstack generated 

by the Stack Creator macro and a sub-directory containing the ROIs generated by 

the FRET Analysis – Cell Markup macro. 

3. Follow the prompt to define the experimental settings by entering the number of 

baseline/stimulated/positive control time points, and whether the data should be 

output as FRET/donor or donor/FRET (see Note 39). Press OK. 

4. Once the macro has finished the batch analysis, graph the F/F0 values for all ROIs 

to visualise responses from the entire cell population using GraphPad Prism. 

5. To be included in further analysis, cells must meet the following criteria. First, cells 

must demonstrate at least a 5-10% change in F/F0 (F/F0 value greater than 1.05- 

1.10) following addition of the positive control. Second, the change in F/F0 following 

addition of the positive control should not exceed 200% (F/F0 value of 2). 

6. Group F/FMax values for all cells using GraphPad Prism (see Note 40 and Fig.3) 

[Fig 3 near here] 

3.4.5. Stack Creator v2.0 – Macro Function Details 

1. Initial Setup. Values that will be needed later in the macro must first be created 

and set to zero. These values include: the file separator (to avoid issues in moving 

between operating systems), the total number of files to process, the total number 

of baseline frames, the total number of stimulated frames, and the total number of 

positive frames (see Note 32). 

2. Locating Files. The source directories for the files must be defined. The macro will 

ask the user to select the source directories for the baseline, stimulated and positive 

control images, and to select a directory within which to save the resulting 

hyperstacks (see Note 41). 

 
3. Defining Experimental Conditions. To correctly parse all the captured files, the 

user must define the conditions of image capture, including: the total number of 

wells; the number of fields of view per well; which well to process first (see step 4); 

the number of time frames captured for baseline, stimulated and positive control 

time series; whether the first file in the directory list is from the donor channel or the 

FRET channel; and whether an auto-alignment should be performed to correct for 

any drift and cell movement throughout the time series. 

//Macro to merge individual image planes into stacks from three separate directories. 
//directories contain baseline frames, stimulated frames and positive control frames 
 
//store global variables for later use 
//fs for file separator (allows cross compatibility between mac and pc) 
//fn for total file number 
//bc for total baseline frame number 
//sc for total stimulate frame number 
//pc for total positive stimulation frame number 
var fs = File.separator; 
var fn = 0; 
var bc = 0; 
var sc = 0; 
var pc = 0;	

//ask user to select baseline, stimulated, positive and output directory lists 
dir1 = getDirectory("Choose Baseline Source Directory "); 
list1 = getFileList(dir1); 
Array.sort(list1); 
 
 
dir2 = getDirectory("Choose Stimulated Source Directory "); 
list2 = getFileList(dir2); 
Array.sort(list2); 
 
 
dir3 = getDirectory("Choose Positive Source Directory "); 
list3 = getFileList(dir3); 
Array.sort(list3); 
 
 
dir4 = getDirectory("Choose Output Directory"); 

4. Starting at a Designated Well. There are instances when the user may wish to 

analyse a defined subset of wells. This option allows the user to begin processing at 

any well within the directory, ignoring all wells preceding this point during acquisition. 

 
5. Generation of Hyperstacks. The macro automatically processes all files in the 

selected directories. This uses one of two sections of code, depending on whether 

the FRET or donor images are listed first within the directory. The first version 

assumes the donor images are first and is run under the If(donorFirst==1) logic. The 

second version assumes the FRET images are first, and is run under the 

if(donorFirst==0) logic. The value for donorFirst is set by the user in the initial 

configuration dialog by the “Is the Donor Channel First in the File List?” check box 

(see Note 42). To generate the combined hyperstacks, the macro first creates a 

series of blank “dummy” images, into which it then copies each captured image 

within the time series. 

//prompt user for number of frames collected for each set and configure auto alignment 
 
Dialog.create("Stack Creator Settings"); 
Dialog.addNumber("Enter the Total Number of Wells Scanned", 0); 
Dialog.addNumber("Enter the Number of Fields Scanned Per Well", 1); 
Dialog.addNumber("Enter the Well to Start Processing From", 1); 
Dialog.addNumber("Enter the Number of Baseline Frames Captured", 0); 
Dialog.addNumber("Enter Number of Stimulated Frames Captured", 0); 
Dialog.addNumber("Enter Number of Positive Frames Captured", 0); 
Dialog.addCheckbox("Is the Donor Channel First in the File List?", true); 
Dialog.addCheckbox("Perform Auto-alignment?  NOTE - Added processing time and risk of artefacts", 
false) 
 
Dialog.show(); 
numWells = Dialog.getNumber(); 
numFields = Dialog.getNumber(); 
startWell = Dialog.getNumber(); 
baseCapture = Dialog.getNumber(); 
stimulateCapture = Dialog.getNumber(); 
positiveCapture = Dialog.getNumber(); 
donorFirst = Dialog.getCheckbox(); 
doAlign = Dialog.getCheckbox(); 
 
numWells = numWells*numFields; 

	

//if starting from a well other than the first one, other time values need to be adjusted 
appropriately 
 
if (startWell>1){ 
 bc = (startWell-1)*baseCapture;  
 sc = (startWell-1)*stimulateCapture;  
 pc = (startWell-1)*positiveCapture;  
 fn = startWell*(baseCapture*2); 
} 

	



	

 17	

 
6. Batch Processing the Hyperstacks. The baseline images for the donor channel 

are processed first. This part of the code is looped depending on the number of 

baseline images that were entered in the initial configuration dialog. The end result 

is a stack of baseline images for the donor channel with one slice per time point. 

 
This is repeated for the stimulated and positive control donor channel images, then 

the whole process is repeated for the FRET channel images. 

 

7. Correcting for Incorrect name Sorting. Loading files through the Array 

commands puts them in a non-natural order. If there are no leading zeros on single 

digit numbers they will be loaded in the order 

1,10,11,12,13,14,15,16,17,18,19,2,20,21 for example. To correct this the created 

stacks are reordered to be in the correct order. The code example here will only 

process up to 30 timepoints. This code is only applied to the stimulate capture 

series as it is the only one that has enough timepoints to have issues. 

//loop for all images in all directories 
for (i=startWell-1; i<numWells; i++){ 
 setBatchMode(true); 
 
//the following code is for analysing files that are sorted with the donor first 
 if(donorFirst==1){ 
  saveName = list1[fn]; 
  fi=fn; 
 
  //create dummy images to copy the relevant images into 
  newImage("Baseline Stack - Donor", "16-bit black", 2048, 2048, 1); 
  newImage("Stimulate Stack - Donor", "16-bit black", 2048, 2048, 1); 
  newImage("Positive Stack - Donor", "16-bit black", 2048, 2048, 1); 
  newImage("Baseline Stack - FRET", "16-bit black", 2048, 2048, 1); 
  newImage("Stimulate Stack - FRET", "16-bit black", 2048, 2048, 1); 
  newImage("Positive Stack - FRET", "16-bit black", 2048, 2048, 1); 

	

//create stack of baseline captures of donor channel 
  for (j=0; j<baseCapture; j++){ 
   filename = dir1 + list1[bc]; 
   if (endsWith(filename, ".tif")){ 
    open(filename); 
    run("Copy"); 
    selectWindow("Baseline Stack - Donor"); 
    run("Add Slice"); 
    run("Paste"); 
    close(list1[bc]); 
    fn=fn+1; 
    bc=bc+1; 
   }} 
  selectWindow("Baseline Stack - Donor"); 
  setSlice(1); 
  run("Delete Slice"); 

//correct order of time points of the Donor Stack if greater than 9 captured 
   if (stimulateCapture>=10&&stimulateCapture<=19){ 
  
    stimulateEnd=stimulateCapture-10; 
    sub1Start=1; 
    sub1End=1; 
    sub2Start=2; 
    sub2End=sub2Start+stimulateEnd; 
    sub3Start=sub2End+1; 
    sub3End=stimulateCapture; 
     
    selectWindow("Stimulate Stack - Donor"); 
    run("Make Substack...", "  slices="+sub1Start+"-"+sub1End); 
    rename("Sub 1"); 
     
    selectWindow("Stimulate Stack - Donor"); 
    run("Make Substack...", "  slices="+sub2Start+"-"+sub2End); 
    rename("Sub 2"); 
     
    selectWindow("Stimulate Stack - Donor"); 
    run("Make Substack...", "  slices="+sub3Start+"-"+sub3End); 
    rename("Sub 3"); 
     
    run("Concatenate...", "  title=[Concatenated Stacks] image1=[Sub 
1] image2=[Sub 3] image3=[Sub 2]"); 
    selectWindow("Stimulate Stack - Donor"); 
    close(); 
    selectWindow("Concatenated Stacks"); 
    rename("Stimulate Stack - Donor"); 
   } 
    
    
   if (stimulateCapture>=20&&stimulateCapture<=29){ 
    
    stimulateEnd=stimulateCapture-20; 
    sub1Start=1; 
    sub1End=1; 
    sub2Start=2; 
    sub2End=11; 
    sub3Start=12; 
    sub3End=12; 
    sub4Start=13; 
    sub4End=sub4Start+stimulateEnd; 
    sub5Start=sub4End+1; 
    sub5End=stimulateCapture; 
    
    selectWindow("Stimulate Stack - Donor"); 
    run("Make Substack...", "  slices="+sub1Start+"-"+sub1End); 
    rename("Sub 1"); 
     
    selectWindow("Stimulate Stack - Donor"); 
    run("Make Substack...", "  slices="+sub2Start+"-"+sub2End); 
    rename("Sub 2"); 
     
    selectWindow("Stimulate Stack - Donor"); 
    run("Make Substack...", "  slices="+sub3Start+"-"+sub3End); 
    rename("Sub 3"); 
     
    selectWindow("Stimulate Stack - Donor"); 
    run("Make Substack...", "  slices="+sub4Start+"-"+sub4End); 
    rename("Sub 4"); 
     
    selectWindow("Stimulate Stack - Donor"); 
    run("Make Substack...", "  slices="+sub5Start+"-"+sub5End); 
    rename("Sub 5"); 
     
    run("Concatenate...", "  title=[Concatenated Stacks] image1=[Sub 
1] image2=[Sub 3] image3=[Sub 5] image4=[Sub 2] image5=[Sub 4]"); 
    selectWindow("Stimulate Stack - Donor"); 
    close(); 
    selectWindow("Concatenated Stacks"); 
    rename("Stimulate Stack - Donor"); 
   }	

8. Merging Donor and FRET Channel Stacks. The donor and FRET channel stacks 

are merged into a single hyperstack with channel 1 representing the donor stack 

and channel 2 representing the FRET stack (see Note 43). If selected in the initial 

configuration, an automatic alignment using a translation model in the StackReg 

plugin is performed. The alignment can compensate for movement of cells or slight 

inaccuracies in the return of the well by the high content imager throughout the time 

series (see Note 44). 

 
9. Naming Files. The name of the final hyperstack is generated from the first 

complete file name in the baseline directory (e.g. A -1(fld 1 wv CFP - CFP- time 1 - 

0 ms)). Final hyperstacks are saved by well name and field number (e.g. A - 1 - fld 

1). 

 
10. Saving the Final Hyperstacks. The final hyperstack is saved into the output 

directory selected by the user. 

//create one stack incorporating baseline, stimulate and positive stacks of donor images 
 run("Concatenate...", "  title=[Concatenated Stacks] image1=[Baseline Stack - Donor] 
image2=[Stimulate Stack - Donor] image3=[Positive Stack - Donor] image4=[-- None --]"); 
 selectWindow("Concatenated Stacks"); 
 rename("Donor Stack"); 
 
//create one stack incorporating baseline, stimulate and positive stacks of FRET images 
 run("Concatenate...", "  title=[Concatenated Stacks] image1=[Baseline Stack - FRET] 
image2=[Stimulate Stack - FRET] image3=[Positive Stack - FRET] image4=[-- None --]"); 
 selectWindow("Concatenated Stacks"); 
 rename("FRET Stack"); 
 
//if auto-alignment was selected carry it out 
 if(doAlign==1){ 
  selectWindow("Composite"); 
  run("Hyperstack to Stack"); 

run("StackReg", "transformation=Translation"); 
run("Stack to Hyperstack...", "order=xyczt(default) channels=2 

slices="+(baseCapture+stimulateCapture+positiveCapture)+" frames=1 display=Color"); 
  } 
 
//merge donor and fret channels into a single hyperstack 
 run("Merge Channels...", "c1=[Donor Stack] c2=[FRET Stack] create"); 

	

//determine well name from file name 
 forName = indexOf(list1[fi],"("); 
 if (forName==6){ 
  wellName = substring(list1[fi], 0, 6); 

fieldName = substring(list1[fi], 11, 12); 
wellID = wellName+" – fld"+fieldName; 

  } 
 if (forName==5){ 
  wellName = substring(list1[fi], 0, 5); 

fieldName = substring(list1[fi], 10, 11); 
wellID = wellName+" – fld"+fieldName; 

  } 

11. Garbage Collection. Once the final hyperstack is saved, the macro frees all 

memory used by FIJI (see Note 45). 

 

3.4.6. FRET Analysis – Cell Markup v1.0 – Macro Function Details 

1. Initial Setup. The file separator character must be defined to maintain compatibility 

between operating systems. 

 
2. Locating Files. The source directory for the files must be defined. This directory 

must contain the hyperstacks that were generated using the Stack Creator macro. 

The macro also creates a directory in which to save the user defined background 

and cell regions of interest (ROI) within the selected source directory (see Note 41). 

 

3. Looping for all Files. The code that defines the background and cell ROIs is 

looped for all stacks in the source folder. 

 

//save final hyperstack into the user defined output directory 
 saveAs("Tiff", dir4+fs+wellID); 
   
//close all open images 
 run("Close All"); 

	

//run java garbage collection to free up used memory 
 wait(2000); 
 call("java.lang.System.gc"); 
 wait(2000); 
 } 
 
	

//Macro to manually mark out cells for automated analysis of FRET Ratios 
 
//fs for file separator (allows cross compatibility between mac and pc) 
fs=File.separator; 

	

//ask user to select source directory that contains stacks generated by the stack creator macro 
dir=getDirectory("Select Source Directory"); 
list=getFileList(dir); 
Array.sort(list); 
 
 
//create directory to store cell ROIs 
roidir=dir+fs+"ROIs"; 
File.makeDirectory(roidir); 

	

//loop code for all tif images in the selected directory 
for(r=0;r<list.length;r++){ 
 
 fileName=dir+list[r]; 
 
//only process tif images 
 if (endsWith(fileName, ".tif")){ 
  open(fileName); 
  nameStore=getTitle(); 
  run("Select None"); 

4. Set Up for User Input. The macro first clears any existing ROIs from the image, 

and adjusts the contrast to optimise the fluorescent signal. 

 
5. Defining the Background. The background fluorescence intensity in both the 

donor channel and FRET channel must be measured. The user is asked to define a 

region of background (containing no cells) and the background ROI is saved in the 

ROI directory. 

 
6. Defining Cells of Interest. The user is asked to outline any cells required for 

subsequent analysis by the FRET Analysis - Batch Analyse macro. Using the 

freehand tool, draw around each cell to be analysed and add to the ROI Table (see 

Notes 36-38). These ROIs are saved into the ROIs folder

 

//check for any existing ROIs and delete them 
  anyROIs=roiManager("count"); 
   if(anyROIs>0){ 
    roiManager("Deselect"); 
    roiManager("Delete"); 
   } 
 
//auto contrast both the donor and FRET channels to make visualisation easier 
  Stack.setChannel(1); 
  run("Enhance Contrast", "saturated=0.30"); 
  Stack.setChannel(2); 
  run("Enhance Contrast", "saturated=0.30"); 

	

//select the freehand tool and ask the user to define the background 
  setTool("Freehand"); 
  waitForUser("Draw a region on the background"); 
  roiManager("Add"); 
   
//background ROI is saved into the ROIs directory under the image name 
  roiManager("Save", roidir+fs+nameStore+" - Background.zip"); 
   
  roiManager("Deselect"); 
  roiManager("Delete"); 
  run("Select None"); 

	

//select the freehand tool and ask the user to outline all the cells to be analysed 
  setTool("Freehand"); 
  roiManager("Show all with labels"); 
  waitForUser("Mark out Cells of Interest\nPress the T Key to add them to the ROI 
Manager\n \nPress OK when complete"); 
   
//marked cell ROIs are saved into the ROI direcotry 
  roiManager("Save", roidir+fs+nameStore+" - Cells.zip"); 
     
  roiManager("Deselect"); 
  roiManager("Delete"); 
  run("Select None"); 
 
  close(nameStore); 
}} 

3.4.7. FRET Analysis – Batch Analyse v1.0 – Macro Function Details 

1. Initial Setup. The file separator character must be defined to maintain compatibility 

between operating systems. 

//Macro to batch analyse the FRET Ratio, F/F0 and F/FMax values of cells marked out 
//with the FRET Analysis - Cell Markup macro 
 
//fs for file separator (allows cross compatibility between mac and pc) 
fs=File.separator; 
 

2. Locating Files. The source directory for the files must be defined. This directory 

must contain the hyperstacks that were generated using the Stack Creator macro, 

and a sub-directory containing the ROIs generated using the FRET Analysis – Cell 

Markup macro. The macro also creates a directory in which to save the overview 

images of the analysed cells within the selected source directory (see Note 41). 

//ask the user to select the source image directory 
dir=getDirectory("Select Source Directory"); 
list=getFileList(dir); 
Array.sort(list); 
 
 
//locate the ROI directory and create an overview directory to save overview images to 
roidir=dir+fs+"ROIs"; 
overviewSave=dir+fs+"Overview"; 
File.makeDirectory(overviewSave); 
 

3. Defining Experimental Conditions. To correctly analyse the data, the user must 

define the number of time points captured for baseline, stimulated and positive 

control time series, and whether the FRET ratio should be calculated as 

donor/FRET or FRET/donor (see Note 39). 

//create a dialog box to collect capture conditions to aid in calculations 
//also ask user to select the type of FRET calculation 
Dialog.create("FRET Analysis Settings"); 
 
Dialog.addNumber("Enter the Number of Baseline Frames Captured", 0); 
Dialog.addNumber("Enter Number of Stimulated Frames Captured", 0); 
Dialog.addNumber("Enter Number of Positive Frames Captured", 0); 
items = newArray("Donor/FRET", "FRET/Donor"); 
Dialog.addRadioButtonGroup("Analysis Type", items, 2, 1, "Donor/FRET"); 
 
 
Dialog.show(); 
 
baseCapture = Dialog.getNumber(); 
stimulateCapture = Dialog.getNumber(); 
positiveCapture = Dialog.getNumber(); 
analysisType = Dialog.getRadioButton(); 
totalTime=baseCapture+stimulateCapture+positiveCapture; 

4. Data Logging. A customised table will be created for the analysed data with the 

following column titles: image name, cell number, time point, background intensity 

donor, background intensity FRET, raw donor intensity, raw FRET intensity, 

corrected donor intensity, corrected FRET intensity, FRET ratio, background ratio, 

F/F0, F/FMax. 

//define measurements required - only mean intensity is needed 
run("Set Measurements...", "mean redirect=None decimal=3"); 
 
//create a custom table to log the data into 
tableTitle="[FRET Analysis]"; 
 
if (isOpen("FRET Analysis")){ 
 print("Table already open");} 
  
 else{ 
   
  run("Table...", "name="+tableTitle+" width=1500 height=250"); 
print(tableTitle, "\\Headings:Image Name\tCell Number\tTime Point\tBackground Intensity Donor\tBackground 
Intesnsity FRET\tRaw Donor Intensity\tRaw FRET Intensity\tCorrected Donor Intensity\tCorrected FRET 
Intenisty\tFRET Ratio\tBackground Ratio\tF/F0\tF/FMax"); 
 } 
 

5. Batch Analysis. The code will be looped for all files in the selected directory. At the 

start of each loop the hyperstack is opened and the cell ROIs are loaded to count 

the number of cells to be analysed. 

//enter batch mode to speed up processing time 
setBatchMode(true); 
 
//batch process for all files in the selected directory 
for(r=0;r<list.length;r++){ 
  
 fileName=dir+list[r]; 
  
//only process tif files 
 if (endsWith(fileName, ".tif")){ 
   open(fileName); 
   nameStore=getTitle(); 
   run("Select None"); 
    
 //clear any ROIs if present 
   anyROIs=roiManager("count"); 
     
   if(anyROIs>0){ 
    roiManager("Deselect"); 
    roiManager("Delete"); 
   } 
 //load the selected cells ROIs to count them and determine how many cells to loop for 
   roiManager("Open", roidir+fs+nameStore+" - Cells.zip"); 
   numberCells=roiManager("Count"); 
 
   roiManager("Deselect"); 
   roiManager("Delete"); 
   run("Select None"); 
 

6. Calculation of FRET ratios. Once the number of cells to be analysed is defined, 

each cell is processed using two very similar sets of code. The first allows 

calculation of the average baseline and positive control FRET ratios. The second 

can then calculate the F/F0 (FRET ratio / average baseline FRET ratio) and F/FMax 

values ( [FRET ratio – average baseline FRET ratio] / [average positive control 

FRET ratio – average baseline FRET ratio] ) for each time point. 

//loop for the number of cells selected 
   for(i=0; i<numberCells; i++){ 
     
//rezero the averageBaselineFRET and averagePositiveFRET values 
    averageBaselineFRET=0; 
    averagePositiveFRET=0; 
 
//loop for all time points 
    for(j=0; j<totalTime; j++){ 
 
//load the background ROI and measure its intensity in the donor and FRET channels 
     roiManager("Open",roidir+fs+nameStore+" - Background.zip"); 
     roiManager("Select", 0); 
     Stack.setSlice(j+1); 
     Stack.setChannel(1); 
     run("Measure"); 
     donorBackgroundInt=getResult("Mean"); 
     Stack.setChannel(2); 
     run("Measure"); 
     FRETBackgroundInt=getResult("Mean"); 
//clear background ROI 

     roiManager("Deselect"); 
     roiManager("Delete"); 
     run("Select None"); 
 
    //load the cells ROIs and select one for processing. 
     roiManager("Open",roidir+fs+nameStore+" - Cells.zip"); 
     roiManager("Select", i); 
       
    //select the correct timepoint to measure 
     Stack.setSlice(j+1); 
       
    //measure the raw intensity of the donor and FRET channels 
     Stack.setChannel(1); 
     run("Measure"); 
     rawDonorInt=getResult("Mean"); 
     Stack.setChannel(2); 
     run("Measure"); 
     rawFRETInt=getResult("Mean"); 
 
     //subtract the background values from donor and FRET values 
     correctedDonorInt=rawDonorInt-donorBackgroundInt; 
     correctedFRETInt=rawFRETInt-FRETBackgroundInt; 
    //calculate Corrected FRET Ratio based on analysis type selected by user 
     if(analysisType=="Donor/FRET"){ 
      FRETRatio=correctedDonorInt/correctedFRETInt; 
     } 
 
     if(analysisType=="FRET/Donor"){ 
       FRETRatio=correctedFRETInt/correctedDonorInt; 
     } 
      
    //calculate the background ratio between the channels 
     backgroundRatio=donorBackgroundInt/FRETBackgroundInt; 
 
   //if at the correct time point (baseline or positive) store the FRET ratio for future calculation 
     if((j+1)<=baseCapture){ 
      averageBaselineFRET=averageBaselineFRET+FRETRatio; 
     } 
      
     if((j+1)>(baseCapture+stimulateCapture)){ 
     averagePositiveFRET=averagePositiveFRET+FRETRatio; 
     } 
    //advance to the next time point in preparation for the next loop 

     Stack.setSlice(j+2); 
 
   //clear any ROIs 
     roiManager("Deselect"); 
     roiManager("Delete"); 
     run("Select None"); 
     } 
 
 //before looping again calculate the average Baseline and Positive FRET ratios 
    averageBaselineFRET=averageBaselineFRET/baseCapture; 

    averagePositiveFRET=averagePositiveFRET/positiveCapture; 

 //loop for all cells calculating F0 and FMax as average baseline and positive FRET ratios are known 
    for(j=0; j<totalTime; j++){ 
      

 The second set of code allows for calculation of F/F0 and F/FMax. 
 
 //calculate the F0 and FMax ratio values 
     fZero=FRETRatio/averageBaselineFRET; 
     fMax1=FRETRatio-averageBaselineFRET; 
     fMax2=averagePositiveFRET-averageBaselineFRET; 
     fMax=fMax1/fMax2; 
 
 //calculate the background intensity ratio 
     backgroundRatio=donorBackgroundInt/FRETBackgroundInt; 
      
 //advance to the next time point in preparation for the next loop 
     Stack.setSlice(j+2); 
 

7. Logging Data. The calculated data is logged into the customised table. 
 //log all measured and calculated values out to the table 
     print(tableTitle, 
nameStore+"\t"+(i+1)+"\t"+(j+1)+"\t"+donorBackgroundInt+"\t"+FRETBackgroundInt+"\t"+rawDonorInt+"\t"+rawF
RETInt+"\t"+correctedDonorInt+"\t"+correctedFRETInt+"\t"+FRETRatio+"\t"+backgroundRatio+"\t"+fZero+"\t"+f
Max); 
 //clear any ROIs 
     roiManager("Deselect"); 
     roiManager("Delete"); 
     run("Select None"); 
     } 
     
 //clear the results table for the next run 
     run("Clear Results"); 
     }} 
 

8. Overview images. To keep a record of which cells were analysed, the first time 

frame of the FRET channel is copied and the cell ROIs are loaded. This image is 

then flattened and saved as a RGB tiff into the Overview directory. 

 //create a copy of the first time point FRET image and auto contrast it 
 run("Duplicate...", "title=Temp duplicate channels=2 slices=1"); 
 run("Enhance Contrast", "saturated=0.30"); 
  
 //load the cell ROIs onto it, flatten the image and save it out to the overview directory 
 roiManager("Open",roidir+fs+nameStore+" - Cells.zip"); 
 roiManager("Show all with labels"); 
 run("Flatten"); 
 saveAs("Tiff", overviewSave+fs+nameStore+" - Overview"); 
 close(nameStore); 
 close("Temp"); 
 

9. Garbage Collection. The macro frees memory used by FIJI (see Note 45). 
//run java garbage collection to free up used memory 
 wait(2000); 
 call("java.lang.System.gc"); 
 wait(2000); 

 } 

10. Saving Logged Data. The custom table is saved as an excel file into the source 

directory and all other tables are closed. 

//save the completed table out to the selected source directory as an xls file 
selectWindow("FRET Analysis"); 
saveAs("Text", dir+fs+"FRET Analysis.xls"); 
 
//close the custom and results tables 
selectWindow("FRET Analysis"); 
run("Close"); 
selectWindow("Results"); 
run("Close"); 

4. Notes 

1. Addition of Penicillin/Streptomycin antibiotics to DMEM, 5% FCS after transfection 

is recommended. 

2. For primary neuronal cultures, antimycotic solution is recommended (25 μg/mL 

Amphotericin B). 

3. Most cell lines adhere well to PDL-coated plates. However, primary neuronal 

cultures require PDL and laminin-coated plates. Once plates are coated with PDL, 

incubate with 50 μL of 50 μg/mL laminin for the duration of the DRG isolation 

protocol. Remove laminin and air-dry plates after DRG trituration, and prior to 

nucleofection (see Step 11 “3.1.2. Dorsal root ganglion neurons isolation and 

culture”). 

4. Using a Bunsen burner, insert the tip of a glass Pasteur pipette into the flame and 

rotate. Monitor the diameter of the tip opening, which will be reduced over time. 

Make pipettes with small, medium or large diameter bores. These will be used for 

sequential trituration of suspensions with the aim to yield single cell suspensions. 

5. Coating of plates with PDL can be performed in bulk. Extra plates can be stored at 

4°C for 3-4 weeks. 

6. PEI stock solution requires agitation, heating (60-70°C) and neutral pH (pH 7) to 

dissolve. Adjust pH using 1 M HCl or 1 M NaOH. 

7. Individual cells exhibit large variation in basal activity and in their relative capacity to 

maximally induce a signalling mediator. Therefore, in addition to measuring, for 

example, the response to receptor activation, the baseline and maximal response 

capacity must also be determined for every cell. Different stimuli are used to 

determine the maximal response capacity of a cell, subject to the signalling pathway 

under investigation (see Table 2). 

8. This protocol is focused on the use of the specified high content imager. However, 

any high content imager can be used, provided that appropriate adjustments to 

FRET detection protocols are performed. 

9. DRG isolation may take up to 3.5 hr: 1 hr to prepare reagents (coating plates, 

preparation of digestion buffer and fire-polishing of pipettes), 20-40 min to isolate 

DRGs of one mouse, 1 hr of enzymatic and mechanical digestion and 30 min for 

nucleofection and plating. 

10. Our protocol is based on adult C57BL/6J mice, 6-8 weeks, male or female. 

11. Ribs can be used as landmarks if specific spinal levels are required. 

12. This step can be performed by pinning down the vertebral column onto a silicone 

elastomer-lined dish and immersing the column in cold CMF-HBSS.  

13. The total amount of DNA used will depend on cell type, and whether cells are to be 

transfected with FRET biosensor alone or co-transfected with receptor/protein of 

interest and FRET biosensor. For most model cell lines (HEK293, CHO, COS, 

HeLa), good expression is achieved using 90 ng/well FRET biosensor alone, or 55 

ng/well receptor/protein of interest and 40 ng/well FRET biosensor. For MCF7 cells, 

good expression is achieved using 200 ng/well FRET biosensor, or 100 ng/well 

receptor/protein of interest and 200 ng/well FRET biosensor. 

14. A modified transfection protocol can be used for cells that are difficult to transfect, 

such as the MCF7 cell line. In this case, cells are transfected and seeded into the 

96-well plate at the same time, to increase penetrance of the transfection mix. The 

DNA:PEI solution is incubated at room temperature for 10-30 min, cells are diluted 

in DMEM with 5% FCS and insulin to achieve 70% confluency, then mixed with the 

DNA:PEI solution immediately prior to seeding. Each well has a final volume of 220 

μL (200 μL cell dilution, 20 μL DNA:PEI solution). Experiments are performed 48 h 

after transfection and seeding. 

15. It is important to keep the final concentration of NaCl relatively low so as not to 

disturb the osmolarity of the cell growth medium. 

16. Serum restriction medium for MCF7 cells should contain 10 μg/mL insulin. 

17. Significant variation in activation can be observed between individual cells within a 

population (see Introduction). As such each cell is considered a ‘unique’ repeat, and 

only one well is collected per experimental condition within each biological repeat. 

Experiments should be repeated with at least 2-3 independent transfections on 

different days. 

18. The nucleofection solution is cytotoxic after extended periods of time. 

19. For DRG cultures, two washes with HBSS are recommended to remove cellular 

debris. 

20. Re-check pH of HBSS prior to use. Standard fluorescent proteins (e.g. YFP) are 

very sensitive to changes in pH, and addition of acidic solutions will alter the FRET 

ratio irrespective of FRET biosensor activation. Experiments in systems that 

experience local changes in pH (e.g. activity induced changes in local calcium 

concentrations in neurons) will need to use modified FRET sensors with pH-

insensitive fluorescent proteins [10, 11]. 

21. It is critical to choose the correct plate, as the recorded dimensions dictate the 

autofocus of the instrument. If a plate is not listed, manually enter plate dimensions 

(available from many Supplier’s websites) into the Plate/Slide Manager and/or use 

the Laser Autofocus Trace option to check plate parameters. 

22. The 40x objective should be suitable for most experiments. If transfection efficiency 

is low or if the cells are large (i.e. DRG neurons, MCF7 cells) either the 20x or 10x 

objectives can be used. Aim to capture at least 50 cells per field of view. 

23. The INCell 2000 does not support simultaneous imaging of two emission channels 

following a single excitation. As images are captured sequentially, it is important to 

capture the FRET image prior to capture of the donor image. Other high content 

imagers do support simultaneous detection of two emission channels (e.g. Perkin 

Elmer Opera). 

24. Exposure times will change depending on the objective selected, and the 

transfection efficiency of the FRET biosensors. These settings should be optimised 

at the commencement of each experiment (see Section 3.3.3). 

25. In the INCell 2000, the polychroic filters included in the FRET module are optimised 

for the following filter pairs: Quad3 for CFP, YFP, dsRed, Cy5; and Quad4 for DAPI, 

FITC, dsRed and Cy5. 

26. A Time Series can be further customised by deselecting individual time points or 

wavelengths. For example, while a reference brightfield image is useful, acquisition 

of this wavelength throughout the entire experiment is unnecessary and extends the 

acquisition time of every well. 

27. Within a single well, multiple fields of view can be acquired as defined within the 

Plate/Slide Viewer window. This is especially useful when acquiring FRET data 

from DRG neurons or cells with low transfection efficiency. 

28. The number of wells that can be imaged within a single time series will depend on 

exposure time, well layout, number of fields of view per well, and the interval of the 

time series. If wells are in close proximity, a 1 min interval will allow capture of up to 

15 wells, whereas a 20 sec interval will allow capture of up to 5 wells. For all high 

content imagers, a motorised stage is required to capture more than one well in a 

given run. The temporal resolution will be limited by how fast this stage can 

(accurately) move between wells.  

29. Ensure the image is not saturated by checking the %Optimum value below the 

image (this should not exceed ~80%). Exposure times should be as short as 

possible to minimise the exposure to excitation light that can induce phototoxic 

damage. The balance between exposure time and intensity of emitted light will be 

governed by several factors: i) Detectors: more sensitive detection systems will 

record more emitted light and therefore allow shorter exposure times; and ii) Filters, 

shutters and light sources: these processes may take up to an extra 500 ms to 

close/turn off following the designated exposure time. So while the camera 

acquisition may be short, the length of time the cells are exposed to light may be 

much longer. Faster termination of the excitation source can be achieved through 

the use of filter wheels, fast shutters or LED light sources. 

30. The INCell 2000 takes approximately 20 seconds to start an image acquisition. To 

minimise the time between drug addition and acquisition of the first image, eject the 

plate with the ‘Stimulate’ Acquisition Protocol window open (and the Base Image 

Folder for image stack storage pre-defined), add drugs simultaneously from a 

compound plate using a multi-channel pipette, gently replace plate and ‘Run 

Protocol’. 

31. The field of view, focus and exposure settings must be identical for the acquisition 

of all three time series (baseline, stimulated and positive control), to allow for the 

calculation of FRET ratios in the final, merged time series following image alignment 

(see Section 3.4.1). 

32. The raw macro code is presented in the format it appears in the FIJI script editor. 

Different colours represent the different functions of the code: Green indicates 

comments and notes that are for information purposes only and do not run as code; 

Blue indicates global system commands; Yellow indicates FIJI/ImageJ commands; 

and Magenta indicates text strings for names and command configuration. 

33. The macros will work in the standard FIJI or ImageJ distribution (ImageJ may need 

the StackReg plugin installed). Computer performance does not limit the macros. A 

standard data set (e.g. total of 28 time points) requires less than 2GB of RAM. CPU 

speed will only affect the speed at which the macros are completed. 

34. This macro is designed for data captured using a GE INCell 2000 high content 

imager in the format of one donor channel and one FRET channel in three sets – 

baseline, stimulated and positive control. It requires all files (baseline, stimulated 

and positive control) to be in separate folders. The macro is adaptable to different 

numbers of time points, wells and fields of view. 

35. Enter “1” to start processing from the first well in the time series list, enter “2” to 

start processing from the second well in the time series list (this will ignore well 1), 

enter “3” to start processing from the third well in the time series list (this will ignore 

wells 1 and 2), and so forth. 

36. Check that the images in the time series are aligned. If the stack is not well aligned, 

manually use an alternative plugin that allows selection of a defined image region to 

use during the alignment (e.g. cvMatch Template plugin available at 

https://sites.google.com/site/qingzongtseng/template-matching-ij-plugin). 

37. Only select cells with the appropriate FRET biosensor localisation and good 

morphology. Include within the selection cells with low levels of fluorescence; the 

FRET ratio change within these cells is generally much greater than cells with high 

fluorescence (these cells typically have increased levels of non-specific FRET). 

While it may be tempting to select the cell with a high and bright signal, in most 

cases larger FRET changes are usually obtained from a cell with a mid to low level 

of expression. 

38. When analysing primary neuronal cultures, it is likely that only a subset of neurons 

will endogenously express the protein of interest. To identify specific neuronal 

subtypes, staining of neurons using neurochemical markers (e.g. isolectin B4, 

CGRP, neurofilament 200) upon completion of the experiment is recommended. 

Subsequent ROI selection and FRET analysis of only positively stained cells will 

more accurately reflect the response of the selected population. 

39. Depending on the biosensor the FRET ratio can be calculated as either 

‘FRET/donor’ or the inverse ‘donor/FRET’ where donor = average intensity of donor 

fluorophore upon donor excitation, and FRET = average intensity of acceptor upon 

donor excitation. For example, increased ERK or Rac1 activity causes an increase 

in FRET within the EKAR and the RaichuEV-Rac1 biosensors; calculating the 

FRET/donor ratio represents this increase. In contrast, increases in cAMP or PKC 

activity cause a decrease in FRET within the Epac2-camps or the CKAR 

biosensors; calculating the inverse donor/FRET ratio allows a corresponding 

graphical representation of this increase. 

40. It is important to consider that individual cells may display distinct temporal profiles 

within the same well. It is therefore recommended to plot and analyse the 

responses at a single cell level before plotting the overall ‘cell population’ response 

detected in a single well (this overall response will be a composite of the individual 

responses within the well). Such analysis can deliver important information about 

the signalling processes under consideration (Fig. 2).  

41. The Array.sort command ensures correct file order across all operating systems. 

42. If the donor channel is not listed first in the source directories, the macro runs a 

nearly identical code instead. In this alternative code, the macro builds the FRET 

channel stack first and then the donor channel stack. However, the final combined 

stack is still in the format of channel 1 = donor, channel 2 = FRET. 

43. A single hyperstack makes all subsequent data management and analysis easier. It 

avoids the need to store and handle individual images (i.e. for each time point) or 

multiple stacks (i.e. one stack per channel) for each experiment. 

44. The automatic alignment adds extra processing time and requires more RAM. The 

plugin can sometimes generate erroneous data by aligning different cells (e.g. if the 

image contains few cells, or moving cells that are highly fluorescent). Check the 

alignment of images within the hyperstack carefully. If the stack is not well aligned, 

manually use an alternative plugin that allows selection of a defined image region to 

use during the alignment (e.g. cvMatch Template plugin available at 

https://sites.google.com/site/qingzongtseng/template-matching-ij-plugin). 

45. Image processing in FIJI is carried out in the Java environment. Images that are 

opened and subsequently closed do not always release the memory they used. 

This is not a problem when the data set is small, but can result in errors/crashing 

with large data sets. 

Tables 

Table 1. List of FRET biosensors used in this protocol 

Targe
t 

Name FRET pair Localisation Lab References Addgen
e 

Plasmid 

cAMP Epac2-camps CFP/YFP 
cytosol 

 
MJ Lohse [7] 

 
[12, 13] 

 

 

plasma 
membrane 

DMF Cooper  

ERK EKAR Cerulean/Venus 
GFP/RFP 

cytosol K Svoboda 
[6] 
[14] 

18679 
18680 

nucleus K Svoboda 18681 
18682 

PKC CKAR CFP/YFP 

cytosol 
 

AC Newton 

[5] 

14860 

plasma 
membrane 

AC Newton 14862 

Rac1 RaichuEV-Rac1 CFP/YFP Rac1 
localisation 

M Matsuda [15, 16] 
 

 

 

Table 2. List of positive controls 

Target FRET Biosensor Positive Control 

cAMP Epac2-camps 10 μM forskolin 
100 μM 3-isobutyl-1-methylxanthine (IBMX) 

ERK EKAR 200 nM phorbol 12,13-dibutyrate (PDBu) 
PKC CKAR 200 nM PDBu 

1x phosphatase inhibitor cocktail 
Rac1 RaichuEV-Rac1 1 μM isoprenaline 

50 ng/mL epidermal growth factor (EGF) 
AlF4- (10 μM AlCl3, 10 mM NaF) [17]  

 

Table 3. Microscopy settings for CFP/YFP and GFP/RFP FRET biosensors 

 

FRET Pair Wavelength Polychroic Order Excitation filter Emission filter Exposure 
CFP/YFP 1 Brightfield DAPI 0.3 

Quad3  2 CFP YFP 0.1 
 3 CFP CFP 0.1 
GFP/RFP 1 Brightfield DAPI 0.3 

Quad4  2 FITC dsRed 0.1 
 3 FITC FITC 0.1 

Figure Captions 

Figure 1: Subcellular targeting of FRET biosensors expressed in model cell lines 

and primary neuronal cultures. (A) Plasma membrane targeted cAMP biosensor 

(pmEpac2-camps) expressed in HEK293 cells, (B) nuclear ERK biosensor (nucEKAR) 

expressed in HEK293 cells, (C) cytosolic ERK biosensor (cytoEKAR) expressed in dorsal 

root ganglia neurons and (D) Rac 1 biosensor (RaicuEV-Rac1) expressed in MCF7 cells. 

Scale bars represent 10 µm (A, B) or 100 µm (C, D). 

Figure 2: Single cell analysis allows discrimination between different cell 

populations depending on signalling kinetics. A) within the same image, cells can elicit 

different responses (e.g. transient versus sustained), that cannot be detected when using 

traditional population assays (B). 

Figure 3: Workflow of the steps to calculate FRET ratios from images. A) Stacking 

images in one file, B) determine the region of interest (ROI) and background, C) plot F/Fo 

for each individual cell and D) group F/FMax values for all cells. 

 

 

 

  


